Memoirs of the School of Engineering, Okayama University Vol.13, February 1979

Computer Program for Numerical Evaluation of the work

Junichi USHIO* , Hirokazu OSAKI* ,
and Susumu KIKUCHI*

(Received February 6, 1979)

Synopsis

We try to evaluate the manual work numerically from a point of view of the homogeneousness and the simultaneousness of both hands using the results of the micromotion study.

The weighted coefficient and the balance index are used to evaluate the homogeneousness of both hands. And the simultaneous index is used to evaluate the simultaneous movement of both hands. It is necessary to make a program in order to use efficiently the method to calculate the indexes or the coefficients.

Therefore the computer program of these methods is mentioned in this paper.

1. Introduction

The motion study is a method to analyze the manual work in detail [1]. However the aim of this method is to visualize the sequence of the manual work by using five symbols or the therbligs.

Therefore we usually use this method to show the process of the work graphically and to evaluate the contents of the work. And though many principles to improve or design the work are mentioned in the principle of the motion economy, there isn't criterion to evaluate the work quantitatively.

---

* Department of Industrial Science

Then we propose a method to evaluate the work quantitatively from the frequency or the elapsed time of the therblig analyzed by the micromotion study [1].

But as this method needs to classify the data many times and to calculate the various values to the criterions, it is necessary to develop the computer program. In this program we can obtained results of not only the proposed method but also the basic analysis of the micromotion study.

2. Method of Numerical Evaluatior of the Work

2.1 Micromotion Study

The data from the micromotion study are basic ones for the proposed method to evaluate the manual work numerically.

In the micromotion study the manual work is divided into therbligs of each hand. Further the end time of each therblig is measured. The observed therblig and its end time are basic data for our method. When the results are given as the input data of the program, all the calculations can be done by computer.

Although the sort of the therbligs is usually seventeen, we use eleven therbligs because six therbligs hardly appear in usual manual work. But we add three therbligs, that is, return, pass shift, and movement.

The result of the micromotion study of the work is shown in simo chart, Table 1.

2.2 Numerical Evaluation Method

A work is evaluated numerically from a point of view of the homogeneousness and the simultaneousness of the therbligs acted with each hand.

The index of the homogeneousness of the work shows how the same kind of therblig is acted with each hand. The index of simultaneousness of the work shows how the same kind of therblig with each hand at the same time.

The frequency and the elapsed time of each therblig of each hand are cumulated from the results of the micromotion study. And the frequency of the simultaneous therblig of both hands is cumulated into l(left hand) $\times$ m(right hand) contingency table [3].

These are shown in Table 2, Table 3, and Table 4.

Table 1. Simo chart

Left hand			Right hand		
therblig symbol	therblig code	end time (1/100 sec)	therblig symbol	therblig code	end time (1/100 sec)
↷	9	↓	∪	1	3346
↓	↓	3351	∩	2	3351
∪	1	3381	#	4	3359
∩	2	3391	⋓	3	↓
⋓	3	3421	↓	↓	3421
9	6	3451	9	6	3451
#	4	3486	#	4	3486
⌓	5	3501	⌓	5	3501
∪	1	3566	↷	9	↓
∩	2	3573	↓	↓	3746
⋓	3	↓	∪	1	3766
↓	↓	3778	∩	2	3778
△	12	3810	⋓	3	↓
⋓	3	3868	↓	↓	3868
9	6	3913	9	6	3913
U	7	4345	U	7	4345
⌓	5	4351	⌓	5	4351
↷	9	4408	∪	1	4386
∪	1	4423	∩	2	4393
∩	2	4445	⋓	3	4430
Ω	8	4918	U	7	↓
⌓	5	4933	↓	↓	4933
↷	9	↓	⋓	3	5021
↓	↓	5031	⌓	5	5031

↓ ; the continuation of the same therblig

Table 2. The result of micromotion study

frequency

THERBLIG	∪ 1	∩ 2	⋓ 3	...	Ω j	...	# m	total
Right hand	n_{11}	n_{12}	n_{13}	...	n_{1j}	...	n_{1m}	N_1
Left hand	n_{21}	n_{22}	n_{23}	...	n_{2j}	...	n_{2m}	N_2

time

	1	2	3	...	j	...	m	total
Right hand	t_{11}	t_{12}	t_{13}	...	t_{1j}	...	t_{1m}	T_1
Left hand	t_{21}	t_{22}	t_{23}	...	t_{2j}	...	t_{2m}	T_2

Table 3. The percentage of the frequency and the elapsed time

percentage (frequency) $a_{ij} = n_{ij} / N_i$

THERBLIG	∪ 1	∩ 2	⊍ 3	… …	Ω j	… …	⧺ m	total
Right hand	a_{11}	a_{12}	a_{13}	…	a_{1j}	…	a_{1m}	1
Left hand	a_{21}	a_{22}	a_{23}	…	a_{2j}	…	a_{2m}	1

percentage (time) $b_{ij} = t_{ij} / T_i$

	1	2	3	…	j	…	m	total
Right hand	b_{11}	b_{12}	b_{13}	…	b_{1j}	…	b_{1m}	1
Left hand	b_{21}	b_{22}	b_{23}	…	b_{2j}	…	b_{2m}	1

Table 4. Contingency table

L \ R		∪ 1	∩ 2	… …	⧺ m	total
∪	1	n_{11}	n_{12}	…	n_{1m}	$n_{1.}$
∩	2	n_{21}	n_{22}	…	n_{2m}	$n_{2.}$
⋮	⋮	⋮	⋮	…	⋮	⋮
↗o	l	n_{l1}	n_{l2}	…	n_{lm}	$n_{l.}$
total		$n_{.1}$	$n_{.2}$	…	$n_{.m}$	n

2.2.1 Index of the homogeneousness of the work

[1] Weighted coefficient (W.C.)

As from the difference of the frequency of each hand in each therblig the homogeneousness can be evaluated, we define the weighted coefficient as its index [2]. The weighted coefficient;W_j is calculated from the equations (1), (2).

$$\sigma_j^2 = (n_{1j} + n_{2j}) - (n_{1j} + n_{2j})^2 / (N_1 + N_2) \quad \ldots\ldots (1)$$

$$W_j = 100 \times (a_{1j} - a_{2j}) / \sqrt{\sigma_j^2} \qquad j = 1, 2, \cdots, m \quad \ldots\ldots (2)$$

[2] Balance index (B.I.)

As (a_{1j}, a_{2j}) or (b_{1j}, b_{2j}), $j=1,2,\cdots,m$ are represented as the points in two dimensional plane, from the degree of scattering the homogeneousness can be evaluated. Then we define the balance index[3]. The balance index; SA, SB are calculated from the equations (3), (4), (5), (6).

$$A_j = (a_{1j} - a_{2j})/\sqrt{2} \quad j=1,2,\cdots,m \quad \cdots\cdots (3)$$

$$SA = \sqrt{\{\sum_{j=1}^{m} A_j^2 - (\sum_{j=1}^{m} A_j)^2/m\}/m} \quad \cdots\cdots (4)$$

$$B_j = (b_{1j} - b_{2j})/\sqrt{2} \quad j=1,2,\cdots,m \quad \cdots\cdots (5)$$

$$SB = \sqrt{\{\sum_{j=1}^{m} B_j^2 - (\sum_{j=1}^{m} B_j)^2/m\}/m} \quad \cdots\cdots (6)$$

2.2.1 Index of the simultaneousness of the work

[3] Simultaneous index (S.I.)

As from the frequency distribution of therbligs of both hands (Table 4) the simultaneousness of the motions can be evaluated, we define the simultaneous index [4]. The simultaneous index; C is calculated from the equation (7), (8), (9) on the basis of Table 4.

$$C_1 = \sqrt{\chi_0^2 / \{n \times \max(l-1, m-1)\}} \quad \cdots\cdots (7)$$

$$C_2 = \sqrt{\chi_0^2 / \{n \times \min(l-1, m-1)\}} \quad \cdots\cdots (8)$$

$$C = \sqrt{C_1 \times C_2} \quad \cdots\cdots (9)$$

$$\chi_0^2 = \sum_{i=1}^{l} \sum_{j=1}^{m} (n_{ij} - m_{ij})^2/m_{ij}, \quad m_{ij} = \frac{n_{i\cdot} \times n_{\cdot j}}{n} \quad \left[\begin{array}{l} i=1,2,\cdots,l \\ j=1,2,\cdots,m \end{array} \right.$$

3. Program

These procedures are programmed in Fortran IV and the program is the form of subroutine.

The subroutine name is EVHOSI.

SUBROUTINE EVHOSI (W,FT,NR,IA,A,NL,IB,B)

3.1 Argument List

ARGUMENT	I/O	TYPE	SIZE	DEFINITION	
W	OUTPUT	CHARACTER	10	title of the work (10A4)	
FT	INPUT	REAL	1	beginning time of the work	
NR	INPUT	INTEGER	1	number of the therbligs	right-hand
IA	INPUT	INTEGER	500	therblig code	right-hand
A	INPUT	REAL	500	end time of each therblig	right-hand
NL	INPUT	INTEGER	1	number of the therbligs	left-hand
IB	INPUT	INTEGER	500	therblig code	left-hand
B	INPUT	REAL	500	end time of each therblig	left-hand

3.2 Suggestion on Using

3.2.1 W $\leqq$ 40 characters

3.2.2 NR,NL $\leqq$ 500

3.2.3 The therblig code and the end time are stored each in IA(i) and A(i) in order of action of a right hand. And it is the same with a left hand.

3.2.4 Correspondence between the therblig and the therblig code

1-Transport Empty (TE)
2-Grasp (G)
3-Transport Loaded (TL)
4-Assemble (A)
5-Release Load (RL)
6-Position (P)
7-Use (U)
8-Holding (H)
9-Unavoidable Delay (UD)
10-Disassemble (DA)
11-Return* (R)
12-Pass Shift* (PS)
13-Movement* (M)
14-Inspect (I)
15-e.t.c

*;the additional therbligs

3.2.5 Subroutine TITLE, PRINTA, NUMEAL, STADEV, and ASSCOE are used in EVHOSI.

TITLE ; print the title.
PRINTA ; print the results of the micromotion study.
NUMEAL ; compute and print the weithted coefficient.
STADEV ; compute and print the balance index.
ASSCOE ; compute and print the simultaneous index.

The program list is shown in Table 5.

4. Example

The data were results of the micromotion study of a worker in some assembly line.

The title of this data was "A-5 BRAKE ASSEMBLE SRF-NO7".

The given data are shown in Table 6. And the results are shown in Table 7.

References

[1] A.Fujita:"IE no Kiso",Kōgakusha, (1969),p63.

[2] C.Hayashi and T.Murayama:"Shijōchōsa no Keikaku to Jissai", Nikkan Kogyō Shinbunsha, (1965),p295.

[3] H.Ōsaki, S.Kikuchi and M.Ogata:"Computer Aided Statistical Technique", Dōbun Shoin, (1978).

[4] S.Iwahara:"Non Parametric Method", Nippon Bunkasha, (1970), p147.

Table 5. Program Listing

```
      SUBROUTINE EVHCSI(W,FT,NR,IA,A,NL,IB,B)
      DIMENSION IA(500),A(500),IB(500),B(500),W(10),
     1          IHR(15),TR(15),PHR(15),PTR(15),AA(500),
     2          IHL(15),TL(15),PHL(15),PTL(15),BB(500),IR(15
      AA(1)=A(1)-FT
      DO 10 I=2,NR
      J=I-1
      AA(I)=A(I)-A(J)
   10 CONTINUE
      BB(1)=B(1)-FT
      DO 20 I=2,NL
      J=I-1
      BB(I)=B(I)-B(J)
   20 CONTINUE
      DO 90 I=1,15
      IHR(I)=0
      IHL(I)=0
      TR(I)=0.0
      TL(I)=0.0
   90 CONTINUE
      DO 100 I=1,NR
      IF(IA(I).EQ.0) GO TO 100
      KK=IA(I)
      IHR(KK)=IHR(KK)+1
      TR(KK)=TR(KK)+AA(I)
  100 CONTINUE
      DO 200 I=1,NL
      IF(IB(I).EQ.0) GO TO 200
      KK=IB(I)
      IHL(KK)=IHL(KK)+1
      TL(KK)=TL(KK)+BB(I)
  200 CONTINUE
      II=0
      DO 300 I=1,15
      IF(IHR(I).EQ.0.AND.IHL(I).EQ.0) GO TO 300
      II=II+1
      IR(II)=I
      IHR(II)=IHR(I)
      TR(II)=TR(I)
      IHL(II)=IHL(I)
      TL(II)=TL(I)
  300 CONTINUE
      ITHR=0
      ITHL=0
      TTR=0.0
      TTL=0.0
      DO 30 J=1,II
      ITHR=ITHR+IHR(J)
      ITHL=ITHL+IHL(J)
      TTR=TTR+TR(J)
      TTL=TTL+TL(J)
   30 CONTINUE
```

```
      DO 40 I=1,II
      PHR(I)=FLOAT(IHR(I))/FLOAT(ITHR)*100.0
      PHL(I)=FLOAT(IHL(I))/FLOAT(ITHL)*100.0
      PTR(I)=TR(I)/TTR*100.0
      PTL(I)=TL(I)/TTL*100.0
   40 CONTINUE
      CALL TITLE(W)
      CALL PRINTA(ITHR,ITHL,IHR,IHL,TTR,TTL,TR,TL,PHR,PHL,PTR,
     1             PTL,IR,II)
      CALL TITLE(W)
      CALL NUMEAL(ITHR,ITHL,IHR,IHL,PHR,PHL,II,IR)
      CALL STADEV(PHR,PHL,PTR,PTL,II)
      CALL TITLE(W)
      CALL ASSCOE(NR,NL,IA,A,IB,B,IHR,IHL,II,IR)
      RETURN
      END
```

```
      SUBROUTINE PRINTA(ITHR,ITHL,IHR,IHL,TTR,TTL,TR,TL,PHR,PHL,
     1 PTR,PTL,IR,II)
      DIMENSION IHR(15),IHL(15),TR(15),TL(15),
     1          PHR(15),PHL(15),PTR(15),PTL(15),IR(15)
      WRITE (6,600)
      WRITE (6,601) (IR(I),I=1,II)
      WRITE (6,602) ITHR,(IHR(I),I=1,II)
      WRITE (6,603) ITHL,(IHL(I),I=1,II)
      WRITE (6,604)
      WRITE (6,605) (IR(I),I=1,II)
      WRITE (6,606) TTR,(TR(I),I=1,II)
      WRITE (6,607) TTL,(TL(I),I=1,II)
      WRITE (6,608)
      WRITE (6,609) (IR(I),I=1,II)
      WRITE (6,610) (PHR(I),I=1,II)
      WRITE (6,611) (PHL(I),I=1,II)
      WRITE (6,612)
      WRITE (6,613) (IR(I),I=1,II)
      WRITE (6,614) (PTR(I),I=1,II)
      WRITE (6,615) (PTL(I),I=1,II)
  600 FORMAT (1H ,/,40X,'***** FREQUENCY *****')
  601 FORMAT  (1H ,/,37X,'TOTAL',15(I4,1X))
  602 FORMAT (1H ,/,30X,'R-HAND ',I4,1X,15(I4,1X))
  603 FORMAT (1H ,/,30X,'L-HAND ',I4,1X,15(I4,1X))
  604 FORMAT  (1H ,/,40X,'***** TIME *****')
  605 FORMAT (1H ,/,37X,'TOTAL  ',15(I5,2X))
  606 FORMAT (1H ,/,30X,'R-HAND ',F6.2,1X,15F7.2)
  607 FORMAT (1H ,/,30X,'L-HAND ',F6.2,1X,15F7.2)
  608 FORMAT (1H ,/,40X,'***** PERCENTAGE (FREQUENCY) *****')
  609 FORMAT (1H ,/,37X,15(2X,I2,2X))
  610 FORMAT (1H ,/,30X,'R-HAND ',15(F5.1,1X))
  611 FORMAT (1H ,/,30X,'L-HAND ',15(F5.1,1X))
  612 FORMAT (1H ,/,40X,'***** PERCENTAGE (TIME) *****')
  613 FORMAT (1H ,/,37X,15(2X,I2,2X))
  614 FORMAT (1H ,/,30X,'R-HAND ',15(F5.1,1X))
  615 FORMAT (1H ,/,30X,'L-HAND ',15(F5.1,1X))
      RETURN
      END
```

```
      SUBROUTINE TITLE(W)
      DIMENSION W(10)
      WRITE (6,600) (W(I),I=1,10)
  600 FORMAT(1H1,//,39X,10A4,///,
     140X,'1-TRANSPORT EMPTY 2-GRASP  3-TRANSPORT LOAD  4-ASSEMBLE',/,
     240X,'5-RELEASE LOAD    6-POSITION   7-USE          8-HOLDING',/,
     340X,'9-UNAVOIDABLE DELAY 10-DISASSEMBLE 11-RETURN 12-PASS SHIFT',
     4/,40X,'13-MOVEMENT  14-INSPECT  15-E.T.C')
      RETURN
      END
```

```
      SUBROUTINE NUMEAL(ITHR,ITHL,IHR,IHL,PHR,PHL,II,IR)
      DIMENSION IHR(15),IHL(15),PHR(15),PHL(15),IR(15)
      WRITE (6,600)
      TAB=ITHR+ITHL
      DO 10 I=1,II
      AB=IHR(I)+IHL(I)
      X=PHR(I)-PHL(I)
      Y=AB-AB**2/TAB
      WWW=X/SQRT(Y)
      WRITE (6,601) IR(I),WWW
  600 FORMAT (1H ,///,40X,'***** WEIGHTED COEFFICIENT *****',///,
     1      43X,'THERBLIG',4X,'W',//)
  601 FORMAT (1H ,46X,I2,5X,F6.2)
   10 CONTINUE
      RETURN
      END
```

```
      SUBROUTINE STADEV(PHR,PHL,PTR,PTL,II)
      DIMENSION PHR(15),PHL(15),PTR(15),PTL(15)
      TAH=0.0
      TAT=0.0
      AH2=0.0
      AT2=0.0
      DO 10 I=1,II
      AH=(PHR(I)-PHL(I))/1.41421356
      AT=(PTR(I)-PTL(I))/1.41421356
      TAH=TAH+AH
      TAT=TAT+AT
      AH2=AH2+AH**2
      AT2=AT2+AT**2
   10 CONTINUE
      SA=(AH2-TAH**2/FLOAT(II))/FLOAT(II)
      SB=(AT2-TAT**2/FLOAT(II))/FLOAT(II)
      SA=SQRT(SA)
      SB=SQRT(SB)
      WRITE (6,600) SA,SB
  600 FORMAT (1H ,/////,40X,'***** BALANCE INDEX *****',///,
     1      49X,'SA=',F5.2,//,49X,'SB=',F5.2)
      RETURN
      END
```

```
      SUBROUTINE ASSCOE(NR,NL,IA,A,IB,B,IHR,IHL,II,IR)
      DIMENSION IHR(15),IHL(15),A(500),B(500),
     1          IA(500),IB(500),IR(15)
      DIMENSION X(15,15),AM(15,15),AT(15),BT(15),IRR(15),IRL(15)
      M=0
      L=0
      DO 80 I=1,II
      IF(IHR(I).EQ.0) GO TO 80
      M=M+1
      IRR(M)=IR(I)
   80 CONTINUE
      DO 90 I=1,II
      IF(IHL(I).EQ.0) GO TO 90
      L=L+1
      IRL(L)=IR(I)
   90 CONTINUE
      DO 100 I=1,L
      DO 100 J=1,M
      X(I,J)=0.0
  100 CONTINUE
      III=1
      JJJ=1
 1000 CONTINUE
      E=B(III)-A(JJJ)
      IF(E) 200,300,400
  200 IF(E.GE.-0.05) GO TO 300
      K=IB(III)
      N=IA(JJJ)
      DO 110 I=1,L
      IF(IRL(I).NE.K) GO TO 110
      K=I
      GO TO 111
  110 CONTINUE
  111 CONTINUE
      DO 120 I=1,M
      IF(IRR(I).NE.N) GO TO 120
      N=I
      GO TO 121
  120 CONTINUE
  121 CONTINUE
      X(K,N)=X(K,N)+1.
      III=III+1
      GO TO 1000
  400 IF(E.LE.0.05) GO TO 300
      K=IB(III)
      N=IA(JJJ)
      DO 130 I=1,L
      IF (IRL(I).NE.K) GO TO 130
      K=I
      GO TO 131
  130 CONTINUE
  131 CONTINUE
      DO 140 I=1,M
      IF(IRR(I).NE.N) GO TO 140
      N=I
      GO TO 141
  140 CONTINUE
  141 CONTINUE
```

```
      X(K,N)=X(K,N)+1.
      JJJ=JJJ+1
      GO TO 1000
  300 K=IB(III)
      N=IA(JJJ)
      DO 150 I=1,L
      IF(IRL(I).NE.K) GO TO 150
      K=I
      GO TO 151
  150 CONTINUE
  151 CONTINUE
      DO 160 I=1,M
      IF(IRR(I).NE.N) GO TO 160
      N=I
      GO TO 161
  160 CONTINUE
  161 CONTINUE
      X(K,N)=X(K,N)+1.
      IF(III.EQ.NL.AND.JJJ.EQ.NR) GO TO 500
      III=III+1
      JJJ=JJJ+1
      GO TO 1000
  500 CONTINUE
      TT=0.0
      ZZ=0.0
      DO 20 I=1,L
      A1=0.0
      DO 21 J=1,M
      A1=A1+X(I,J)
   21 CONTINUE
      AT(I)=A1
   20 CONTINUE
      DO 30 J=1,M
      A2=0.0
      DO 31 I=1,L
      A2=A2+X(I,J)
   31 CONTINUE
      BT(J)=A2
   30 CONTINUE
      DO 40 I=1,L
      TT=TT+AT(I)
   40 CONTINUE
      DO 50 I=1,L
      DO 50 J=1,M
      AM(I,J)=AT(I)*BT(J)/TT
      Z=(X(I,J)-AM(I,J))**2/AM(I,J)
      ZZ=ZZ+Z
   50 CONTINUE
      C1=SQRT(ZZ/(TT*FLOAT(L-1)))
      C2=SQRT(ZZ/(TT*FLOAT(M-1)))
      C=SQRT(C1*C2)
      LL=L-1
      MM=M-1
      WRITE (6,600)
      WRITE (6,601) (IRR(I),I=1,M)
      WRITE (6,602)
      DO 60 I=1,L
      WRITE (6,603) IRL(I),AT(I),(X(I,J),J=1,M)
   60 CONTINUE
      WRITE (6,604) TT,(BT(I),I=1,M)
      WRITE (6,605) ZZ,C1,C2,C,LL,MM
```

```
600 FORMAT(1H ,///,40X,'*** CONTINGENCY TABLE & ',
   1      'SIMULTANEOUS COEFFICIENT ***',/)
601 FORMAT (1H ,/,40X,'THERBLIG  ',15(I2,2X))
602 FORMAT (1H ,/,44X,'TOTAL')
603 FORMAT (1H ,39X,I2,2X,F4.0,1X,15F4.0)
604 FORMAT (1H ,/,44X,F4.0,1X,15F4.0)
605 FORMAT(1H ,//,45X,'X**2',7X,'C1',8X,'C2',9X,'C',//,40X,
   1F10.3,3(3X,F7.3),//,45X,'DEGREE OF FREEDOM',
   24X,'=',5X,'( ',I2,', ',I2,')')
    RETURN
    END
```

Table 6. Given Data

Title of the work A-5 BRAKE-ASSEMBLE SRF-NO7

Beginning time of the work FT = 33.24 (sec)

Number of therbligs NR = 20 (right hand)
NL = 21 (left hand)

Given data

Left hand		Right hand	
IB(i)	B(i)	IA(i)	A(i)
9	3351	1	3346
1	3381	2	3351
2	3391	4	3359
3	3421	3	3421
6	3451	6	3451
4	3486	4	3486
5	3501	5	3501
1	3566	9	3746
2	3573	1	3766
3	3778	2	3778
12	3810	3	3868
3	3868	6	3913
6	3913	7	4345
7	4345	5	4351
5	4351	1	4386
9	4408	2	4393
1	4423	3	4430
2	4445	7	4933
8	4918	3	5021
5	4933	5	5031
9	5031		

Table 7. Computer Output

***** A-5 BRAKE-ASSEMBLE SRF-NO7 *****

1-TRANSPORT EMPTY 2-GRASP 3-TRANSPORT LOAD 4-ASSEMBLE
5-RELEASE LOAD 6-POSITION 7-USE 8-HOLDING
9-UNAVOIDABLE DELAY 10-DISASSEMBLE 11-RETURN 12-PASS SHIFT
13-MOVEMENT 14-INSPECT 15-E.T.C

***** FREQUENCY *****

	TOTAL	1	2	3	4	5	6	7	8	9	12
R-HAND	20	3	3	4	2	3	2	2	0	1	0
L-HAND	21	3	3	3	1	3	2	1	1	3	1

***** TIME *****

	TOTAL	1	2	3	4	5	6	7	8	9	12
R-HAND	17.07	0.77	0.24	2.77	0.43	0.31	0.75	9.35	0.	2.45	0.
L-HAND	17.07	1.10	0.39	2.93	0.35	0.36	0.75	4.32	4.73	1.82	0.32

***** PERCENTAGE (FREQUENCY) *****

	1	2	3	4	5	6	7	8	9	12
R-HAND	15.0	15.0	20.0	10.0	15.0	10.0	10.0	0.	5.0	0.
L-HAND	14.3	14.3	14.3	4.8	14.3	9.5	4.8	4.8	14.3	4.8

***** PERCENTAGE (TIME) *****

	1	2	3	4	5	6	7	8	9	12
R-HAND	4.5	1.4	16.2	2.5	1.8	4.4	54.8	0.	14.4	0.
L-HAND	6.4	2.3	17.2	2.1	2.1	4.4	25.3	27.7	10.7	1.9

***** A-5 BRAKE-ASSEMBLE SRF-NO7 *****

1-TRANSPORT EMPTY 2-GRASP 3-TRANSPORT LOAD 4-ASSEMBLE
5-RELEASE LOAD 6-POSITION 7-USE 8-HOLDING
9-UNAVOIDABLE DELAY 10-DISASSEMBLE 11-RETURN 12-PASS SHIFT
13-MOVEMENT 14-INSPECT 15-E.T.C

***** WEIGHTED COEFFICIENT *****

THERBLIG	W
1	0.32
2	0.32
3	2.37
4	3.14
5	0.32
6	0.25
7	3.14
8	-4.82
9	-4.89
12	-4.82

***** BALANCE INDEX *****

SA= 3.32

SB= 9.11

***** A-5 BRAKE-ASSEMBLE SRF-NO7 *****

1-TRANSPORT EMPTY 2-GRASP 3-TRANSPORT LOAD 4-ASSEMBLE
5-RELEASE LOAD 6-POSITION 7-USE 8-HOLDING
9-UNAVOIDABLE DELAY 10-DISASSEMBLE 11-RETURN 12-PASS SHIFT
13-MOVEMENT 14-INSPECT 15-E.T.C

*** CONTINGENCY TABLE & SIMULTANEOUS COEFFICIENT ***

THERBLIG	TOTAL	1	2	3	4	5	6	7	9
1	4.	0.	0.	2.	1.	0.	0.	0.	1.
2	4.	0.	0.	2.	0.	0.	0.	1.	1.
3	5.	1.	1.	2.	0.	0.	0.	0.	1.
4	1.	0.	0.	0.	1.	0.	0.	0.	0.
5	3.	0.	0.	0.	0.	2.	0.	1.	0.
6	2.	0.	0.	0.	0.	0.	2.	0.	0.
7	1.	0.	0.	0.	0.	0.	0.	1.	0.
8	1.	0.	0.	0.	0.	0.	0.	1.	0.
9	7.	2.	2.	2.	0.	1.	0.	0.	0.
12	1.	0.	0.	1.	0.	0.	0.	0.	0.
	29.	3.	3.	9.	2.	3.	2.	4.	3.

X**2	C1	C2	C
86.891	0.577	0.654	0.614

DEGREE OF FREEDOM = (9, 7)